


Колоквијум из Практикума из рачунарске анализе кола

Колоквијум се ради **са литературом** у електронској форми и траје 60 минута. Колоквијум се оцењује са 50 поена. Подебљани бројеви у загради на почетку реда представљају број поена додељен делу задатка или питању. Није дозвољено напуштање сале 30 минута од почетка колоквијума. Писати искључиво **хемијском** оловком. Коначне одговоре на питања уписати у одговарајуће правоугаонике или учртати у дијаграме (користити и полеђину). Овај папир се предаје. Попунити податке о кандидату у следећој табелици.

Индекс год./број		Презиме и име					Одсек	
3.1а	3.1б	3.1в	3.1г	3.1д		К.	Σ	Оцена

Предметни наставник: др *Милка Потребих*, ванредни професор


<p>Задатак 1</p> <p>Осцилатор са Виновим мостом (Wien bridge oscillator) има почетну енергију и $a = 3$, $R_1 = R_2 = R$, $C_1 = C_2 = C$.</p> <p>$u_{C_2}(0^-) = U_{02}$</p> 	<p>MATLAB MuPAD</p> <p>3.1а</p> <p>(10) Одредити једначине стања.</p>
<p>3.16</p> <p>(10) Одредити одзив на почетне услове за напон u.</p>	

3.1в

(10) За претходну тачку, нацртати график напона u , у интервалу времена од 0 до 5 ms, за следеће вредности елемената: $R = 10 \text{ k}\Omega$, $C = 16 \text{ nF}$, $U_{02} = 2 \text{ V}$.

MATLAB Simulink Simscape**3.1г&3.1д**

(10) Нацртати шему кола (модел) и (5+5) графике напона $u(t)$ и струје отпорника $i_{R_2}(t)$ ако је побуда троугаони импулс као на слици.




Поставимо систем једначина кола. То је листа једначина.

Знак једнако се користи за куцање једначине, двотачка-једнако је за доделу.

Прве изводе смо писали као нове променљиве Du_{C1} , Du_{C2} .

```
jednaced := [
  iC1 + iR1 + i1 + iR2 = 0,
  iR2 + ig = iC2,
  iC2 = i2 + ig,
  uC1 = R1*iR1,
  uC1 = R2*iR2 + uC2 + u,
  iC1 = C1*DuC1,
  iC2 = C2*DuC2,
  u = 3*uC1, i1 = 0
]
```



```
[i1 + iC1 + iR1 + iR2 = 0, iR2 + ig = iC2, iC2 = i2 + ig, uC1 = R1 iR1, uC1 = u + uC2 + R2 iR2,
 iC1 = C1 DuC1, iC2 = C2 DuC2, u = 3 uC1, i1 = 0]
```

Функција nops (Number of OPerandS) враћа број аргумената и даје број чланова листе.

```
nops(jednaced)
9
```

Уклонити (елиминисати) све променљиве осим променљивих стања.

Променљиве стања су, по дефиницији, напони кондензатора и струје калемова.

Дефинишимо листу напона и струја које треба уклонити.

```
ukloniti := [i1, i2, iC1, iC2, iR1, iR2, u]
[i1, i2, iC1, iC2, iR1, iR2, u]
```

```
nops(ukloniti)
7
```

Користимо функцију за уклањање (елиминацију) променљивих.

Она враћа једначине у којима су само променљиве стања. То су напони кондензатора.

```
jednaceducl2 := groebner::eliminate(jednaced, ukloniti)
[R1 ig - uC1 - C1 DuC1 R1 - C2 DuC2 R1, R2 ig - uC2 - 2 uC1 - C2 DuC2 R2,
 2 R1 uC1 + R1 uC2 - R2 uC1 - C1 DuC1 R1 R2]
```

Једначине које смо добили су изрази који представљају леве стране једнакости.

Десна страна је нула. Дobili смо листу оваквих изрази - једначина.

Дефинишимо замену вредности елемената сходно задатку.

```
zamena := [C1 = C, C2 = C, R1 = R, R2 = R]
[C1 = C, C2 = C, R1 = R, R2 = R]
```

Замена је листа једноставних једначина, раздвојених запетом.

Свака једначина је облика променљива-једнако-вредност.

Једначине стања су једначине по првим изводима променљивих стања написане у договореном облику (Кошијева нормална форма).

Са леве стране једнакости је извод променљиве стања или нула.

Са десне стране је израз од променљивих стања, параметара и побуда.

Решимо систем једначина по променљивама које представљају изводе.

```
resenjeD := linsolve(jednaceducl2, [DuC1, DuC2]) | zamena
[ DuC1 = (R uC1 + R uC2) / (C R^2), DuC2 = - (2 uC1 + uC2 - R ig) / (C R) ]
```

Решење се може, некада, боље сагледати ако се преуреди функцијом expand.

```
expand(resenjeD)
```

$$\left[\left[\text{DuC1} = \frac{uC1}{C R} + \frac{uC2}{C R}, \text{DuC2} = \frac{ig}{C} - \frac{2 uC1}{C R} - \frac{uC2}{C R} \right] \right]$$

Да би решили систем једначина стања треба да прилагодимо синтаксу софтверском алату. Напони и струје треба да буду симболи иза којих је у обичним заградама наведена променљива од које зависе, а то је време.

Први извод се пише као симбол-апостроф-зависност-од-времена.

Побуду замењујемо према захтеву задатка.

Тражимо одзив на почетне услове, у овом задатку, тако да је побуда једнака нули.

$$\left[\begin{array}{l} \text{vreme} := [\\ \quad \text{DuC1} = uC1'(t), \text{DuC2} = uC2'(t), \\ \quad uC1 = uC1(t), uC2 = uC2(t), \\ \quad ig = 0 \\] \\ \left[\text{DuC1} = uC1'(t), \text{DuC2} = uC2'(t), uC1 = uC1(t), uC2 = uC2(t), ig = 0 \right] \end{array} \right]$$

Систем једначина стања је записан као листа две диференцијалне једначине.

$$\left[\begin{array}{l} \text{jednacinestanja} := \text{resenjeD} \mid \text{vreme} \\ \left[uC1'(t) = \frac{R uC1(t) + R uC2(t)}{C R^2}, uC2'(t) = - \frac{2 uC1(t) + uC2(t)}{C R} \right] \end{array} \right]$$

Да би се добио одзив на почетне услове треба додати почетне услове. Почетни напон једног кондензатора је задата, а другог није. Договорно, ако почетни услов није задат, сматрамо да је једнак нули.

Функција за решавање система диференцијалних једначина захтева да се једначине, заједно са почетним условима, задају као скуп, низ једнакости рздвојених запетама и ограђених таласастим заградама.

Променљиве у диференцијалним једначинама се такође задају као скуп.

Једначине стања са почетним условима чине систем једначина - скуп.

$$\left[\begin{array}{l} \text{sistemStanja} := \{ \\ \quad \text{op}(\text{jednacinestanja}), \\ \quad uC1(0) = 0, \\ \quad uC2(0) = U02 \\ \} \\ \left\{ uC2(0) = U02, uC2'(t) = - \frac{2 uC1(t) + uC2(t)}{C R}, uC1(0) = 0, uC1'(t) = \frac{R uC1(t) + R uC2(t)}{C R^2} \right\} \end{array} \right]$$

Низ једначина стања рздвојених запетама, без угластих заграда, добијамо функцијом `op` која враћа списак аргумената (операнде).

$$\left[\begin{array}{l} \text{op}(\text{jednacinestanja}) \\ uC1'(t) = \frac{R uC1(t) + R uC2(t)}{C R^2}, uC2'(t) = - \frac{2 uC1(t) + uC2(t)}{C R} \end{array} \right]$$

Променљиве стања се, такође, наводе као скуп.

$$\left[\begin{array}{l} \text{promenljiveStanja} := \{uC1(t), uC2(t)\} \\ \{uC1(t), uC2(t)\} \end{array} \right]$$

Функција која решава диференцијалне једначине и системе је `ode::solve`.

Решење једначина стања са датим почетним условима је скуп листи.

$$\left[\begin{array}{l} \text{resenjeStanja} := \text{ode}::\text{solve}(\text{sistemStanja}, \text{promenljiveStanja}) \\ \text{assuming}(R > 0 \text{ and } C > 0) \end{array} \right]$$

$$\left\{ \left[u_{C2}(t) = U_{02} \cos\left(\frac{t}{CR}\right) - U_{02} \sin\left(\frac{t}{CR}\right), u_{C1}(t) = U_{02} \sin\left(\frac{t}{CR}\right) \right] \right\}$$

Функција assuming задаје услове конјукцијом and, тако да се задаје домен параметара електричног кола, $R > 0, C > 0$. Без ње, решење може бити приказано као сложенији израз.

$$\text{ode}::\text{solve}(\text{sistemStanja}, \text{promenljiveStanja})$$

$$\left\{ \left[u_{C2}(t) = U_{02} \sigma_1 \left(\frac{1}{2} - \frac{1}{2}i\right) + U_{02} \sigma_2 \left(\frac{1}{2} + \frac{1}{2}i\right), u_{C1}(t) = \frac{U_{02} \sigma_1 i}{2} - \frac{U_{02} \sigma_2 i}{2} \right] \right\}$$

where

$$\sigma_1 = e^{-\frac{t}{CR}}$$

$$\sigma_2 = e^{\frac{t}{CR}}$$

Електрично коло има јединствено решење, тако да је резултат једночлани скуп.

$$\text{odzivStanja} := \text{resenjeStanja}[1]$$

$$\left[u_{C2}(t) = U_{02} \cos\left(\frac{t}{CR}\right) - U_{02} \sin\left(\frac{t}{CR}\right), u_{C1}(t) = U_{02} \sin\left(\frac{t}{CR}\right) \right]$$

Поједина решења добијамо заменом из листе решења.

$$u_{C1Resenje} := u_{C1}(t) \mid \text{odzivStanja}$$

$$U_{02} \sin\left(\frac{t}{CR}\right)$$

$$u_{C2Resenje} := u_{C2}(t) \mid \text{odzivStanja}$$

$$U_{02} \cos\left(\frac{t}{CR}\right) - U_{02} \sin\left(\frac{t}{CR}\right)$$

Решење се може, некада, написати у подеснијем облику општим функцијама за упрошћавање.

$$\text{simplify}(u_{C2Resenje})$$

$$\sqrt{2} U_{02} \cos\left(\frac{\pi}{4} + \frac{t}{CR}\right)$$

$$\text{Simplify}(u_{C2Resenje})$$

$$\sqrt{2} U_{02} \cos\left(\frac{\pi}{4} + \frac{t}{CR}\right)$$

Како да одредимо остале напоне и струје?
Једноставно. Пођимо од полазних једначина.

$$\text{jednaced}$$

$$[i_1 + i_{C1} + i_{R1} + i_{R2} = 0, i_{R2} + i_g = i_{C2}, i_{C2} = i_2 + i_g, u_{C1} = R_1 i_{R1}, u_{C1} = u + u_{C2} + R_2 i_{R2}, i_{C1} = C_1 Du_{C1}, i_{C2} = C_2 Du_{C2}, u = 3 u_{C1}, i_1 = 0]$$


Уклонимо из њих променљиве које означавају прве изводе променљивих стања.

$$\text{ostaleJednacine} := \text{groebner}::\text{eliminate}(\text{jednaced}, [Du_{C1}, Du_{C2}])$$

$$[i_1, i_2 + i_{C1} + i_{R1}, 3 R_1 i_{R1} - u, i_{R2} - i_2, i_{C2} - i_2 - i_g, 2 u + 3 u_{C2} + 3 R_2 i_2, 3 u_{C1} - u]$$

Решимо ове једначине по преосталим променљивама, а то су оне које смо раније уклањали.

$$\text{ostaliOdziv} := \text{linsolve}(\text{ostaleJednacine}, \text{ukloniti})$$



$$\left[\begin{aligned} i_1 = 0, i_2 = -\frac{2u_{C1} + u_{C2}}{R_2}, i_{C1} = \frac{2R_1 u_{C1} + R_1 u_{C2} - R_2 u_{C1}}{R_1 R_2}, i_{C2} = -\frac{2u_{C1} + u_{C2} - R_2 i_g}{R_2}, \\ i_{R1} = \frac{u_{C1}}{R_1}, i_{R2} = -\frac{2u_{C1} + u_{C2}}{R_2}, u = 3u_{C1} \end{aligned} \right]$$

Уочити да су преостале променљиве алгебарски изрази по променљивама стања.
Заменили у њима променљиве стања и побуду.

```
ostaliOdzivResenje := ostaliOdziv |
[uC1 = uC1Resenje, uC2 = uC2Resenje,
ig = 0,
R1 = R, R2 = R, C1 = C, C2 = C
]
```

$$\left[\begin{aligned} i_1 = 0, i_2 = \sigma_1, i_{C1} = \frac{R \left(U_{02} \cos\left(\frac{t}{CR}\right) - \sigma_2 \right) + R U_{02} \sin\left(\frac{t}{CR}\right)}{R^2}, i_{C2} = \sigma_1, i_{R1} = \frac{\sigma_2}{R}, i_{R2} = \sigma_1, \\ u = 3 U_{02} \sin\left(\frac{t}{CR}\right) \end{aligned} \right]$$

where

$$\sigma_1 = -\frac{U_{02} \cos\left(\frac{t}{CR}\right) + \sigma_2}{R}$$

$$\sigma_2 = U_{02} \sin\left(\frac{t}{CR}\right)$$

Тражени напон је

```
uResenje := u | ostaliOdzivResenje
3 U02 sin\left(\frac{t}{CR}\right)
```

Колика је струја iR1?

```
iR1Resenje := iR1 | ostaliOdzivResenje
\frac{U02 sin\left(\frac{t}{CR}\right)}{R}
```

Колика је струја iC2?

```
iC2Resenje := iC2 | ostaliOdzivResenje
-\frac{U02 cos\left(\frac{t}{CR}\right) + U02 sin\left(\frac{t}{CR}\right)}{R}
```

```
Simplify(iC2Resenje)
```

$$-\frac{\sqrt{2} U_{02} \sin\left(\frac{\pi}{4} + \frac{t}{CR}\right)}{R}$$


Нацртати напон u(t),

у интервалу времена од t1 = 0 s до t2 = 5 ms,

за следеће вредности елемената:

C = 16 nF, R = 10 kOhm, U02 = 2 V.

```
vrednostiElementa := [C = 16e-9, R = 10e3, U02 = 2]
[C = 0.000000016, R = 10000.0, U02 = 2]
```




Подесити да Хевисајдова функција буде једнака нула у нули.

```
[ Pref::heavisideAtOrigin(0) :
```

На графику напона $u(t)$ обележити осе. Подесити да се црта у 4000 тачака.

```
plot( uResenje | vrednostiElementa,
t = 0..5e-3, XMesh = 4000,
AxesTitles = ["vreme t [s]", "napon u [V]" ] )
```



Симболима, променљивама, напона и струја нису додељиване вредности зато што их користимо у функцијама за решавање једначина. На пример, u , $u(t)$ је само симбол напона, а $uResenje$ је израз који представља решење.

```
[ u(t) = uResenje
```

$$u(t) = 3 U_{02} \sin\left(\frac{t}{CR}\right)$$

```
[ version()
```

```
[ [8, 1, 0]
```

